Data Repository Queries from NPR

Data Viewer reports for looking at Data Repository data in MAGIC, C/S and 6.x
(Demo & downloads only)
MIS.USER.zcus.is.dr.mat.data.viewer.muse.2017 (6.x) 
MIS.USER.zcus.is.dr.npr.data.viewer.muse.2017 (6.x and C/S) 
MIS.USER.zcus.is.dr.data.viewer.muse.2017 (MAGIC)

e.g. 6.1 NPR data viewer selections

[bookmark: _GoBack]Lookups based on previous selections.  
DISTINCT, TOP N, WHERE, ORDER BY added for extra flexibility, given some SQL expertise, but all are optional.  E.g.
[image: cid:image001.png@01D2D53F.8F7EBF30]
 ‘Element’ is an optional selection – if left blank, all populated elements from selected record will be chosen, up to maximum of 1000 rows or user-supplied limit:

(Test data shown)

LAB.L.SPEC.zcus.is.abnormal.results.with.diagnoses.from.dr  
·  Simple list of abnormal lab results, pulls ABS or REG diagnoses from Data Repository per patient.

Example of diagnoses from ABS (latest effective date retrieved as part of SQL query) – Test data shown :

From REG (no ‘latest effective date’ available in DR data) – Test data shown :

Written from LAB.L.SPEC.tests, LAB.L.SPEC.number.x index.

Select screen : 

[image: ]
[image: ]

(Provide for retrieving Diagnoses from RegAcct or AbsAcct.)


Report picture and line / field attributes:

‘start’ macro (change highlighted parts to match site specifics) :
; link to DR so calls to DR.ODBC.util in 'sk3' (patient sort) macro will work
%Z.link.db(`DR',`O,B,1'),
IF{L(@.ring@Tr.l2u,"TEST")<L(@.ring) "testfdb";"livefdb"}^/DRDB,
@CONNECTION.STRING,
@LOAD.DR.ODBC

CONNECTION.STRING
"Driver={"_IF{@DR.PARAM.odbc.driver;"SQL Server"}_"};"^/CSTRING,
/CSTRING_"Server="_IF{@DR.PARAM.server;"CAL-DR02"}_";"^/CSTRING,
IF{c.xx.user&c.xx.pwd /CSTRING_"UID="_c.xx.user_";Pwd="_c.xx.pwd_";"^/CSTRING;
   /CSTRING_"Trusted Connection=yes;"^/CSTRING},
/CSTRING_"Initial Catalog="_/DRDB_";"^/CSTRING,
; the following is just to identify what process is accessing DR -- arbitrary, but may be logged?
/CSTRING_"App=DataRepository - "_@.ring_" - "_@.hcis_" - CS_SendQuery"^/CSTRING

LOAD.DR.ODBC
; if SQL credentials supplied, use them
; --otherwise, try trusted credentials (no connection string)
; specify B argument to “LoadODBC” to customize buffer size (default is 1024x1024 = 1 MB? )
; specify E argument true to "Connect" call, retains connection string in /HRO["CS"]
IF{%DR.ODBC.util("LoadODBC");
   IF{c.xx.user&c.xx.pwd %DR.ODBC.util("Connect",/CSTRING);
      %DR.ODBC.util("Connect")}}^/ERR


‘sk3’ macro (‘patient’ sort level)  :
; retrieve ABS OR REG diagnoses from Data Repository
patient^PHA.PAT.patient,
c.xx.abs.or.reg^TYPE,
;
""^SUB,
DO{>/QUERY[SUB]^SUB ""^/QUERY[SUB]},
""^/QUERY,
;
IF{TYPE$1@Tr.l2u="A" @ABS.QUERY;
   TYPE$1@Tr.l2u="R" @REG.QUERY},
;
%DR.ODBC.util("Execute",{"QRY",{^/QUERY}}),
%DR.ODBC.util("FetchCursor","",1),
; B argument 'nil' for all rows; otherwise, # of rows to fetch
; C argument TRUE to retrieve column names in row 0
;
; results will be in /HRO["R",...] here until CloseCursor invoked
; (OR error messages might be present instead)
@RESULTS,
%DR.ODBC.util("CloseCursor")

ABS.QUERY
" SELECT ABD.DiagnosisEffectiveDateID"^/QUERY[SUB+1^SUB],
" ,ABD.DiagnosisUrnID"^/QUERY[SUB+1^SUB],
" ,ABD.DiagnosisCode_MisDxID"^/QUERY[SUB+1^SUB],
" ,COALESCE(MDX.Name,ABD.DiagnosisName) AS DiagnosisName"^/QUERY[SUB+1^SUB],
" FROM "_/DRDB_".dbo.AbsAcct_Diagnoses ABD"^/QUERY[SUB+1^SUB],
" LEFT JOIN "_/DRDB_".dbo.MisDx_Main MDX ON (MDX.SourceID = ABD.SourceID AND "^/QUERY[SUB+1^SUB],
" MDX.MisDxID = ABD.DiagnosisCode_MisDxID )"^/QUERY[SUB+1^SUB],
" WHERE ABD.DiagnosisEffectiveDateID = ( SELECT MAX "^/QUERY[SUB+1^SUB],
; SYNTAX CHECKER WILL CHOKE ON "X (" with no space between...
" (ABD1.DiagnosisEffectiveDateID)"^/QUERY[SUB+1^SUB],
" FROM "_/DRDB_".dbo.AbsAcct_Diagnoses ABD1"^/QUERY[SUB+1^SUB],
" WHERE ABD1.SourceID = ABD.SourceID AND ABD1.VisitID = ABD.VisitID ) "^/QUERY[SUB+1^SUB],
" AND ABD.VisitID = '"_PHA.PAT.patient_"' "^/QUERY[SUB+1^SUB],
" ORDER BY ABD.DiagnosisUrnID "^/QUERY[SUB+1^SUB],
" ;"^/QUERY[SUB+1^SUB]

REG.QUERY
; make sure Diagnosis Code and Name are in same columns as above query
; (we'll suppress the 'date' query from above on output if REG is chosen instead)
;
" SELECT RDX.[VisitID]"^/QUERY[SUB+1^SUB],
" ,RDX.[SortOrder]"^/QUERY[SUB+1^SUB],
" ,RDX.[DiagnosisCode_MisDxID]"^/QUERY[SUB+1^SUB],
" ,COALESCE(MDX.[Name],RDX.[DiagnosisAlternateName]) AS DiagnosisName"^/QUERY[SUB+1^SUB],
"  FROM "_/DRDB_".dbo.RegAcct_Diagnoses RDX"^/QUERY[SUB+1^SUB],
"  LEFT JOIN "_/DRDB_".dbo.MisDx_Main MDX ON (MDX.SourceID = RDX.SourceID AND MDX.MisDxID = RDX.DiagnosisCode_MisDxID )"^/QUERY[SUB+1^SUB],
"  WHERE RDX.VisitID  = '"_PHA.PAT.patient_"' "^/QUERY[SUB+1^SUB],
"  ORDER BY RDX.[VisitID], RDX.[SortOrder] "^/QUERY[SUB+1^SUB],
" ; "^/QUERY[SUB+1^SUB]

RESULTS
; see 'doc' macro for explanation of DR.ODBC structures and subscripts
; retrieve results from /HRO["R",...] in more NPR-friendly form.
; row 0 values are column names
;/HRO["R","B",batch,"R",row,col,1]|0 = value 
;  (where batch=1 here, as there's only one statement executed)
; ==>>  /DIAG[ROW,COL] = value, where /DIAG[0,COL] = column name
;
""^SUB,
DO{>/DIAG[SUB]^SUB ""^/DIAG[SUB]},
""^/DIAG,
;
""^DR.ODBC.tmp.row.q^DR.ODBC.tmp.row.col.q,
1^DR.ODBC.tmp.batch.q,
DO{@Next(DR.ODBC.tmp.row.col.q) 1^DR.ODBC.tmp.row.col.value.q,
                                @DR.ODBC.tmp.row.col.value^/DIAG[DR.ODBC.tmp.row.q,DR.ODBC.tmp.row.col.q];
   @Next(DR.ODBC.tmp.row.q)}



Example of results after ABS query executed:

Simplified to:

